
Before We Start

All attendee lines are muted

This is to prevent interruptions during the 
presentation. 

X

The webinar is being recorded

After the webinar, we will send you a mail to give access 
to the recording and presentation slides.

Please submit questions via 
the Chat or Q&A panel

Your questions will be addressed directly during 
the webinar or in the Q&A section after the 
presentation.

Share your feedback with us

Use the link provided at the end to let us know 
what we can do better.

Some Terminology

• We are talking about 32/64 bit applications, not operating systems
– Everyone should be on a 64-bit OS right now

– Platform == bitness of the application, not the OS

• We are talking about Windows
– Not Mac, Linux, iOS, etc. 

– 32-bit == Win32 or x86

– 64-bit == Win64 or x64 or (confusingly) x86_64

• This presentation is as-of Notes 12.0.2 FP3 and Notes 14.0.0
– Some things might change or get fixed later!

– 12.0.2 FP3 just came out on 17 January 2024

Why 64-Bit?

• 32-bit applications still work on 64-bit Windows
– WOW64 redirection under the covers

– HKLM\Software → HKLM\Software\WOW6432Node

– C:\Windows\System32 → C:\Windows\SysWOW64

• 64-bit advantages
– More memory (virtual address space) per process: 128TB vs 4GB

• Not just process memory, also used for memory mapping files (less swapping)

• https://www.panagenda.com/blog/the-out-of-memory-mystery-in-notes-11

– More CPU registers available for use

• Potentially faster

• https://learn.microsoft.com/en-us/windows-hardware/drivers/debugger/x64-architecture

https://www.panagenda.com/blog/the-out-of-memory-mystery-in-notes-11
https://learn.microsoft.com/en-us/windows-hardware/drivers/debugger/x64-architecture


Better Security?

• Larger memory space allows for better address randomization
– ASLR high entropy randomization

– Viruses/exploits can’t guess where your applications are running in memory

• Win32 App Isolation in Windows 11
– https://blogs.windows.com/windowsdeveloper/2023/06/14/public-preview-improve-

win32-app-security-via-app-isolation

• 32-bit supporting libraries/programs not maintained
– Unmaintained software is potentially open to more exploits

https://blogs.windows.com/windowsdeveloper/2023/06/14/public-preview-improve-win32-app-security-via-app-isolation
https://blogs.windows.com/windowsdeveloper/2023/06/14/public-preview-improve-win32-app-security-via-app-isolation


Politics

• Some organizations have 64-bit only directives

• 32-bit programs seem “old”

• Even Microsoft struggles with this with Office
– Still has both 32- and 64-bit versions

– https://support.microsoft.com/en-us/office/choose-between-the-64-bit-or-32-bit-
version-of-office-2dee7807-8f95-4d0c-b5fe-6c6f49b8d261

https://support.microsoft.com/en-us/office/choose-between-the-64-bit-or-32-bit-version-of-office-2dee7807-8f95-4d0c-b5fe-6c6f49b8d261
https://support.microsoft.com/en-us/office/choose-between-the-64-bit-or-32-bit-version-of-office-2dee7807-8f95-4d0c-b5fe-6c6f49b8d261


Better Support Long Term

• Hard to support multiple versions of software
– Both HCL and you!

• Eclipse dropped 32-bit support in 2018
– Although 32-bit Java packages are still being built

• Pain of switching now will be worth it later

• Already did it with Domino servers years ago
– First Windows 64-bit Domino server was version 8.0.1  (2008)

– Last Windows 32-bit Domino server was version 9.0.1  (2013)



Domino vs Notes Client

• Didn’t we figure out all the issues when we switched to 64-bit Domino?

• So many differences…
– UI code

– Eclipse integration

– Third-party libraries and app integration

– Broader range of API calls (both Notes and Windows)

– Users do much crazier things than admins do

• Client has a lot more code running on a much wider variety of machines

How Do I Know?

• How do you know if you’re using a 64-bit client?
– Notes 14: it’s always 64-bit

– Notes 12.0.2: Help – About

– Notes 12.0.1 or earlier: 32-bit

• 12.0.2 is the only version 
with both 32- and 64-bit



64-bit Client: Task Manager

• Task Manager shows “HCL Notes (32-bit)” for 32-bit client
– 64-bit client just says “HCL Notes”



64-bit Client: Registry

• 32-bit client has registry entries in WOW6432Node branch
– Use reg query with /reg:32 or /reg:64 and see which one works

– DANGER: sometimes a machine that used to have a 32-bit client still has the old 
registry entries!



64-bit Client: Check Java Version

• Run “java -version” from Notes\jvm\bin
– Look for 32-Bit or 64-Bit



64-bit Client: Check Java Properties

• Or run:  java -XshowSettings -version 2>&1 | findstr -i os.arch
– Check for “amd64” (64-bit) or “x86” (32-bit)

– Even machines with Intel processors report “amd64” for 64-bit Java



64-Bit Client: Check for x86_64 Eclipse Plugins

• In Notes\framework\rcp\eclipse\plugins 64-bit client has x86_64 
folders



64-bit Client: In Your Code

• LotusScript
– NotesSession.Platform = "Windows/64"

• Formula
– @Platform() = "Windows/64"

• Java
– System.getProperty("os.arch").contains("64") && 

System.getProperty("os.name").contains(“Windows");



BONUS: Notes Client Version from Command Line

• You can check Notes client version (not bitness) from command line
– https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0101159

– nsd -version

– notes.exe -v notesver.txt

https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0101159

HCL Recommendations for 64-Bit

:12



HCL Guidelines For Updating Applications

https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0101520

https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0101520


Guideline 1: 64-bit libraries



Guideline 1: 64-bit libraries

• 64-bit ODBC drivers to external data sources
– Also 64-bit NotesSQL/HCL ODBC driver to access the Notes client

– No longer available (EOM December 2022, EOS June 2024)

– HCL suggests using https://www.cdata.com/drivers/domino/odbc

• Any DLLs used as extension manager addins
– Virus scan, VOIP, MarvelClient!

• NotesPeek
– 64-bit version available on FlexNet

https://www.cdata.com/drivers/domino/odbc


EXTMGR_ADDINS and 32-bit DLLs?

• What happens if you try to load a 32-bit DLL using ExtMgr_Addins?
– Nothing, just a weird 21:C1 error in console.log

– 21:C1 = ERROR_BAD_EXE_FORMAT : not a valid Win32 Application

Guideline 2: COM and OLE Compatibility

• Depends on how the called application handles COM/OLE calls

– Out-of-process COM servers use RPC, can potentially work with any bitness

– Can be designed to accept calls from 32- or 64-bit applications (“AnyCPU” or proxy DLLs)

– Needs registry entries in both the 32- and 64-bit CLSID branches

• https://web.archive.org/web/20190428064632/https://techtalk.gfi.com/32bit-object-64bit-environment

• https://stackoverflow.com/questions/38655598/lotus-domino-on-64-bit-system-could-not-create-
automation-object-error-208

• Try it and see what happens!

– I tried with some simple OLE calls to Microsoft Word…

https://web.archive.org/web/20190428064632/https:/techtalk.gfi.com/32bit-object-64bit-environment
https://stackoverflow.com/questions/38655598/lotus-domino-on-64-bit-system-could-not-create-automation-object-error-208
https://stackoverflow.com/questions/38655598/lotus-domino-on-64-bit-system-could-not-create-automation-object-error-208


Example: Calling Word using OLE

Dim wordApp As Variant
Dim wordDoc As Variant

On Error Resume Next
Set wordApp = GetObject("", "Word.Application")

On Error GoTo processError
If IsEmpty(wordApp) Then

Set wordApp = CreateObject("Word.Application")
End If

If IsEmpty(wordApp) Then
MessageBox "Could not create Word Application"
Exit Sub

End If

wordApp.Visible = True
wordApp.documents.Add
Set wordDoc = wordApp.ActiveDocument

Dim paragraph As Variant
Dim pos As Long
Set paragraph = wordDoc.Paragraphs(wordDoc.Paragraphs.Count).Range
pos = paragraph.End-1
Call paragraph.InsertAfter("Hello Clippy!")

'wordApp.Quit
Exit Sub

processError:
    MessageBox "Error " & Err & " on line " & Erl & ": " & Error

Exit Sub



COM/OLE from Office to Notes

• Use Notes.NotesSession (OLE) not Lotus.NotesSession (COM)
– COM gives Run-time error ‘-2147221164 (80040154)’: Class not registered



Embedded OLE Objects on Documents

• Worked for me with 32-bit Notes client and 64-bit Office 365

• Embedded OLE object functions might have problems on 64-bit Notes
– https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0106900

Command 12.0.2 FP2 12.0.2 FP3 14.0.0

AutoLaunch: First OLE Object No No No

NotesUIDocument.CreateObject No No No

NotesRichTextItem.EmbedObject No No Yes

https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0106900


So… Does 64-bit Notes Require 64-bit Office?

• I don’t know man… it depends?

• Light testing with different bitness:
– 64-bit Word + 32-bit Notes client agent: worked

– 32-bit Word + 64-bit Notes client agent: worked

– Office VBA to Notes worked for me using CreateObject(“Notes.NotesSession”)

– Embedded OLE object on document worked on 32-bit Notes, not always on 64-bit

• Your results may vary, depending on what you’re trying to do

• However, if you need to check the bitness of your installed version of 
Office, here are some ways to do it



Task Manager: Look for “(32-bit)”

• While Word (or another Office program) is running, open Task Manager
– If it says “(32-bit)” after the program 

name, it’s 32-bit

– If there’s nothing after the program 
name, it’s 64-bit



Registry: Check the Path

• Run this from a command prompt or PowerShell: 

reg query "HKLM\SOFTWARE\Microsoft\Windows\CurrentVersion\App Paths\WINWORD.EXE"

– If the path is “C:\Program Files (x86)” it’s almost certainly 32-bit

– If the path is “C:\Program Files” it’s almost certainly 64-bit



LotusScript: Find Bitness of Installed MS Word

Const SCS_32BIT_BINARY = 0 'A 32-bit Windows-based application
Const SCS_DOS_BINARY = 1   'An MS-DOS – based application
Const SCS_WOW_BINARY = 2   'A 16-bit Windows-based application 
Const SCS_PIF_BINARY = 3   'A PIF file that executes an MS-DOS –
based application
Const SCS_POSIX_BINARY = 4 'A POSIX – based application
Const SCS_OS216_BINARY = 5 'A 16-bit OS/2-based application
Const SCS_64BIT_BINARY = 6 'A 64-bit Windows-based application

' only works on EXEs, not DLLs
Declare Function GetBinaryTypeA Lib "kernel32.dll" (_

ByVal lpApplicationName As String, _
lpBinaryType As Long) As Integer

'** find the path to the Word executable, e.g. 
'** C:\Program Files\Microsoft Office\root\Office16\WINWORD.EXE
Dim RegValue As Variant
RegValue = Evaluate(|@RegQueryValue("HKEY_LOCAL_MACHINE"; 
"SOFTWARE\\Microsoft\\Windows\\CurrentVersion\\App 
Paths\\WINWORD.EXE"; "")|)

If (RegValue(0) = "") Then
MessageBox "Registry key not found"
Exit Sub

End If

Dim binaryType As Long
Dim result As Integer

result = GetBinaryTypeA(RegValue(0), binaryType)
If result = 0 Then

MessageBox "Call failed: " & result
ElseIf (binaryType = SCS_32BIT_BINARY) Then

MessageBox "32-bit application detected"
ElseIf (binaryType = SCS_64BIT_BINARY) Then

MessageBox "64-bit application detected"
Else

MessageBox "Unknown application bitness: " & binaryType
End If



Guideline 3: Array of Longs on 64-bit Client



Guideline 3: Array of Longs on 64-bit Client

FIXED in 
12.0.2 FP2 

and Notes 14



Make Your Data Work For You

LotusScript and API Calls

LotusScript on Windows 64-bit clients

• In general, LotusScript works the same on 64-bit
– No functions/classes were removed

– No recompile necessary

• Potential issue with smaller stack size (recursive functions)
– https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0102290

• Max size of static String array is 4049 (was 8190 on 32-bit)
– https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0040725

• A couple functions that can cause problems when compiled on 64-bit
– If you compile on 64-bit and run on 32-bit, they don’t work correctly!

– If you compile on 32-bit, they work properly on both platforms

https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0102290
https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0040725


Compiled on 64-Bit, Broken on 32-Bit

• NotesView.GetAllEntriesByKey
– Specifically when using an array of Variants for a lookup key

– When compiled on 64-bit, causes “Invalid key value type” error when run on 32-bit

– If you use an array of Strings it might return the wrong result with no error!

– https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0102291

Sub Initialize
Dim session As New NotesSession
Dim key(1) As Variant
Dim vec As NotesViewEntryCollection
Dim v As NotesView

key(0) = "Value1"
key(1) = "Value2"

Set v = session.CurrentDatabase.GetView("Sorted")
Set vec = v.GetAllEntriesByKey(pKey, True)

MsgBox "vec.Count = " & vec.Count
End Sub

When compiled on 64-bit

Run on Result

64-bit e.g.: vec.Count = 1

32-bit Error: Invalid key 
value type

https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0102291


Compiled on 64-Bit, Broken on 32-Bit

• Replace(str, arr1, arr2)
– Specifically when using an array of Strings for the search value

– When compiled on 64-bit, always returns “” when run on 32-bit

– There is no error! Fixed in 32-bit 12.0.2 FP3 , broken on all other (older) 32-bit clients

– https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0109692

Sub Initialize
Dim s As String, arr(7) As String
arr(0)="\"
arr(1)="/"
arr(2)=":"
arr(3)="*"
arr(4)="?"
arr(5)="<"
arr(6)=">"
arr(7)="|"
s = "Hello : World"
s = Replace(s, arr, "")
MessageBox s

End Sub

When compiled on 64-bit

Run on Result

64-bit “Hello  World”

32-bit “”

https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0109692


LotusScript Debugger: Buggy in 12.0.2, Better in 14

• Lots of fixes in 14 to solve Debugger problems you saw in 12.0.2
– 12.0.2 FP3 got some (all?) of these fixes too



Platform-Specific Code

• Code with NotesSession.Platform or IsDefined() needs to be reviewed
– Recommendation: use NotesSession.Platform = “Windows/64” for 64-bit code

– NotesSession.Platform = “Windows/32” 

• WILL NOT RECOGNIZE 64-BIT

• 64-bit client is “Windows/64”

– Left(NotesSession.Platform, 3) = “Win”

• THINKS 32- AND 64-BIT ARE THE SAME

– IsDefined(“WIN32”)

• THINKS 32- AND 64-BIT ARE THE SAME 

• IsDefined(“WIN64”) always False even on 64-bit client



C-API Calls

• If a parameter or return value is data, same as 32-bit
– STATUS, WORD, DWORD, int, long, etc.

• If a parameter is a LotusScript String, same as 32-bit

• If a parameter or return value is a Notes C-API HANDLE, same as 32-bit
– Still uses 32-bit sizes for backwards compatibility

• If a parameter or return value is a Windows HANDLE or a pointer…
– Use Long on 32-bit clients, Double on 64-bit clients

– Call session.UseDoubleAsPointer = True before you make the calls on 64-bit

– Call session.UseDoubleAsPointer = False after you make the calls on 64-bit



NotesSession.UseDoubleAsPointer

• NotesSession.UseDoubleAsPointer on 64-bit platforms
– Only required for API calls returning or using pointers/handles on a 64-bit client 

– Set to True before the API code runs, False after it’s done

– https://help.hcltechsw.com/dom_designer/12.0.2/basic/H_USEDOUBLEASPOINTER_METHOD

_SESSION.html

– https://ds-infolib.hcltechsw.com/ldd/ddwiki.nsf/dx/C-Callouts_on_non-32-bit_Platforms

• NotesSession is shared by all code triggered from the client UI
– Buttons, actions, agents…

– Persistent setting across databases while the client is open

– If you set UseDoubleAsPointer in one place, you are setting it everywhere

• Does not seem to include code triggered by NotesAgent.Run

https://help.hcltechsw.com/dom_designer/12.0.2/basic/H_USEDOUBLEASPOINTER_METHOD_SESSION.html
https://help.hcltechsw.com/dom_designer/12.0.2/basic/H_USEDOUBLEASPOINTER_METHOD_SESSION.html
https://ds-infolib.hcltechsw.com/ldd/ddwiki.nsf/dx/C-Callouts_on_non-32-bit_Platforms


NotesSession.UseDoubleAsPointer

• Doesn’t work on Forms in 12.0.2 FP1/2
– E.g. buttons, actions, events…

– Error 156:  “Property is not available on this object”

– Fixed in 12.0.2 FP3 and 14.0.0

• What if you might have to compile the code on a pre-12.0.1 client?
– UseDoubleAsPointer didn’t exist before 12.0.1, won’t compile

– Use an Execute statement… but make sure it only runs on 64-bit clients

– All 64-bit clients are guaranteed to recognize this property

If (session.Platform = "Windows/64") Then
Execute |Dim session As New NotesSession

session.UseDoubleAsPointer = True|



Test All Your C-API Calls

• Avoid using “Any” as a parameter type for pointers when possible
– Can work, but won’t autobox numerical values to the correct type/size

– Use a strict parameter type (Long or Double) whenever possible to avoid surprises

• Some things that worked on 64-bit 12.0.2 require changes for Notes 14!
– Also some things that worked on 12.0.2 FP1 require changes for 12.0.2 FP2/3

– 12.0.2 FP1 silently converted some handles to Long, FP2/Notes 14 does not

– Notes 14 is more strict about NotesSession.UseDoubleAsPointer

• Some things that worked on Notes 14 Beta require changes for GA!
– Same situation as 12.0.2

– You just have to test all the API calls and modify as needed

4:30



Notes C-API Example

Declare Function StatQueryTime Lib "nnotes.dll" (_
ByVal facility As String, _
ByVal statName As String, _
ByVal headerString As String, _
ByVal namePrefix As String, _
ByVal valuePrefix As String, _
ByVal lineSuffix As String, _
dhStats As Long, _                '** DHANDLE to a text buffer
dwStatsLength As Long) As Integer '** returns a STATUS

Declare Function OSLockObject Lib "nnotes.dll" (_
ByVal dHandle As Long) As Long                 '** takes a DHANDLE, returns a pointer

Declare Function OSUnlockObject Lib "nnotes.dll" (_
ByVal dHandle As Long) As Long                 '** takes a DHANDLE, returns a BOOL

Declare Function OSMemFree Lib "nnotes.dll" (_
ByVal dHandle As Long) As Integer              '** takes a DHANDLE, returns a STATUS

Declare Sub CopyBufferToString Lib "nnotes.dll" Alias "Cmovmem" (_
ByVal lpSrc As Long, _      '** pointer to a source buffer
ByVal lpDest As String, _   '** destination string
ByVal lSize As Long)        '** size of destination string



Notes C-API Example

Declare Function StatQueryTime Lib "nnotes.dll" (_
ByVal facility As String, _
ByVal statName As String, _
ByVal headerString As String, _
ByVal namePrefix As String, _
ByVal valuePrefix As String, _
ByVal lineSuffix As String, _
dhStats As Long, _                '** DHANDLE to a text buffer
dwStatsLength As Long) As Integer '** returns a STATUS

Declare Function OSLockObject Lib "nnotes.dll" (_
ByVal dHandle As Long) As Long                 '** takes a DHANDLE, returns a pointer

Declare Function OSUnlockObject Lib "nnotes.dll" (_
ByVal dHandle As Long) As Long                 '** takes a DHANDLE, returns a BOOL

Declare Function OSMemFree Lib "nnotes.dll" (_
ByVal dHandle As Long) As Integer              '** takes a DHANDLE, returns a STATUS

Declare Sub CopyBufferToString Lib "nnotes.dll" Alias "Cmovmem" (_
ByVal lpSrc As Long, _      '** pointer to a source buffer
ByVal lpDest As String, _   '** destination string
ByVal lSize As Long)        '** size of destination string

String, DHANDLE, STATUS, 
and BOOL are fine



Notes C-API Example

Declare Function StatQueryTime Lib "nnotes.dll" (_
ByVal facility As String, _
ByVal statName As String, _
ByVal headerString As String, _
ByVal namePrefix As String, _
ByVal valuePrefix As String, _
ByVal lineSuffix As String, _
dhStats As Long, _                '** DHANDLE to a text buffer
dwStatsLength As Long) As Integer '** returns a STATUS

Declare Function OSLockObject Lib "nnotes.dll" (_
ByVal dHandle As Long) As Long                 '** takes a DHANDLE, returns a pointer

Declare Function OSUnlockObject Lib "nnotes.dll" (_
ByVal dHandle As Long) As Long                 '** takes a DHANDLE, returns a BOOL

Declare Function OSMemFree Lib "nnotes.dll" (_
ByVal dHandle As Long) As Integer              '** takes a DHANDLE, returns a STATUS

Declare Sub CopyBufferToString Lib "nnotes.dll" Alias "Cmovmem" (_
ByVal lpSrc As Long, _      '** pointer to a source buffer
ByVal lpDest As String, _   '** destination string
ByVal lSize As Long)        '** size of destination string

Pointers need to change from 
Long to Double



Notes C-API Example

Declare Function OSLockObject Lib "nnotes.dll" (_
ByVal dHandle As Long) As Long                 '** takes a DHANDLE, returns a pointer

Declare Function OSLockObject_64 Lib "nnotes.dll" Alias "OSLockObject" (_
ByVal dHandle As Long) As Double               '** takes a DHANDLE, returns a pointer

Declare Sub CopyBufferToString Lib "nnotes.dll" Alias "Cmovmem" (_
ByVal lpSrc As Long, _      '** pointer to a source buffer
ByVal lpDest As String, _   '** destination string
ByVal lSize As Long)        '** size of destination string

Declare Sub CopyBufferToString_64 Lib "nnotes.dll" Alias "Cmovmem" (_
ByVal lpSrc As Double, _    '** pointer to a source buffer
ByVal lpDest As String, _   '** destination string
ByVal lSize As Long)        '** size of destination string

Make a copy of each Declare 
(use the Alias keyword) 
with a modified name



Notes C-API Example

Declare Function OSLockObject Lib "nnotes.dll" (_
ByVal dHandle As Long) As Long                 '** takes a DHANDLE, returns a pointer

Declare Function OSLockObject_64 Lib "nnotes.dll" Alias "OSLockObject" (_
ByVal dHandle As Long) As Double               '** takes a DHANDLE, returns a pointer

Declare Sub CopyBufferToString Lib "nnotes.dll" Alias "Cmovmem" (_
ByVal lpSrc As Long, _      '** pointer to a source buffer
ByVal lpDest As String, _   '** destination string
ByVal lSize As Long)        '** size of destination string

Declare Sub CopyBufferToString_64 Lib "nnotes.dll" Alias "Cmovmem" (_
ByVal lpSrc As Double, _    '** pointer to a source buffer
ByVal lpDest As String, _   '** destination string
ByVal lSize As Long)        '** size of destination string

Pointers need to change from 
Long to Double



Notes C-API Example

Dim session As New NotesSession
Dim buffer As Long
Dim bufferLen As Long
Dim bufferText As String
Dim pointer As Long

Call StatQueryTime("Stats", "Time.Start", "", "", Chr(9), Chr(10), buffer, bufferLen)
bufferText = Space(bufferLen)

pointer = OSLockObject(buffer)
Call CopyBufferToString(pointer, bufferText, bufferLen)

Call OSUnlockObject(buffer)
Call OSMemFree(buffer)

MsgBox bufferText



Notes C-API Example

Dim session As New NotesSession
Dim buffer As Long
Dim bufferLen As Long
Dim bufferText As String
Dim pointer As Double

Call StatQueryTime("Stats", "Time.Start", "", "", Chr(9), Chr(10), buffer, bufferLen)
bufferText = Space(bufferLen)

If (session.Platform = "Windows/64") Then
session.UseDoubleAsPointer = True
pointer = OSLockObject_64(buffer)
Call CopyBufferToString_64(pointer, bufferText, bufferLen)
session.UseDoubleAsPointer = False

Else
pointer = OSLockObject(buffer)
Call CopyBufferToString(pointer, bufferText, bufferLen)

End If

Call OSUnlockObject(buffer)
Call OSMemFree(buffer)

MsgBox bufferText



Windows API Example (get clipboard text)

Const CF_UNICODETEXT = 13
Const OS_TRANSLATE_UNICODE_TO_LMBCS = 23

'** 32-bit API calls
Declare Function OpenClipboard Lib "user32.dll" (ByVal hwnd As Long) As Long
Declare Function CloseClipboard Lib "user32.dll" () As Long
Declare Function GetClipboardData Lib "user32.dll" (ByVal wFormat As Long) As Long

Declare Function GlobalLock Lib "kernel32" (ByVal hMem As Long) As Long
Declare Function GlobalUnlock Lib "kernel32" (ByVal hMem As Long) As Long
Declare Function GlobalSize Lib "kernel32" (ByVal hMem As Long) As Long

Declare Function OSTranslateFromPtr Lib "nnotes.dll" Alias "OSTranslate" ( _
ByVal mode As Integer, _
ByVal strIn As Long, _
ByVal lenIn As Integer, _
ByVal strOut As LMBCS String, _
ByVal lenOut As Integer ) As Integer



Windows API Example (get clipboard text)

'** 64-bit API calls 
Declare Function OpenClipboard_64 Lib "user32.dll" Alias "OpenClipboard" (_

ByVal hwnd As Double) As Long               '** hwnd is a window handle
Declare Function GetClipboardData_64 Lib "user32.dll" Alias "GetClipboardData" (_

ByVal wFormat As Long) As Double            '** returns a memory handle
Declare Function CloseClipboard_64 Lib "user32.dll" Alias "CloseClipboard" () As Long

Declare Function GlobalLock_64 Lib "kernel32.dll" Alias "GlobalLock" (_
ByVal hMem As Double) As Double             '** hMem is a memory handle, returns a pointer

Declare Function GlobalUnlock_64 Lib "kernel32.dll" Alias "GlobalUnlock" (_
ByVal hMem As Double) As Long               '** hMem is a memory handle, returns a BOOL

Declare Function GlobalSize_64 Lib "kernel32.dll" Alias "GlobalSize" (_
ByVal hMem As Double) As Long               '** hMem is a memory handle, returns a size

Declare Function OSTranslateFromPtr_64 Lib "nnotes.dll" Alias "OSTranslate" ( _
ByVal mode As Integer, _
ByVal strIn As Double, _                    '** strIn is a string pointer
ByVal lenIn As Integer, _
ByVal strOut As LMBCS String, _
ByVal lenOut As Integer ) As Integer

Same as before, but the window HANDLE and 
memory HANDLEs all became Doubles too



Windows API Example (get clipboard text)
Function GetClipboard() As String

Dim session As New NotesSession
If (session.Platform = "Windows/64") Then

GetClipboard = GetClipboard64()
Exit Function

End If

Dim glbHandle As Long
Dim cbPointer As Long
Dim cbPointerLen As Long
Dim cbString As String

If OpenClipboard(0) Then
glbHandle = GetClipboardData(CF_UNICODETEXT)
cbPointer = GlobalLock(glbHandle)
cbPointerLen = GlobalSize(glbHandle)

cbString = Space(cbPointerLen)
Call OSTranslateFromPtr( OS_TRANSLATE_UNICODE_TO_LMBCS, _

cbPointer, cbPointerLen, cbString, cbPointerLen )
cbString = StrLeft(cbString, Chr(0))

Call GlobalUnlock(glbHandle)
Call CloseClipboard()

End If

GetClipboard = cbString
End Function



Windows API Example (get clipboard text)

Function GetClipboard64() As String
Dim session As New NotesSession
session.UseDoubleAsPointer = True

Dim glbHandle_64 As Double
Dim cbPointer_64 As Double
Dim cbPointerLen As Long
Dim cbString As String

If OpenClipboard_64(0) Then
glbHandle_64 = GetClipboardData_64(CF_UNICODETEXT)
cbPointer_64 = GlobalLock_64(glbHandle_64)
cbPointerLen = GlobalSize_64(glbHandle_64)

cbString = Space(cbPointerLen)
Call OSTranslateFromPtr_64( OS_TRANSLATE_UNICODE_TO_LMBCS, _

cbPointer_64, cbPointerLen, cbString, cbPointerLen )
cbString = StrLeft(cbString, Chr(0))

Call GlobalUnlock_64(glbHandle_64)
Call CloseClipboard_64()

End If

GetClipboard64 = cbString
session.UseDoubleAsPointer = False

End Function



How Do You Know?

• How do you know what’s a handle and what’s a pointer and what’s 
something else?

• API Documentation!
– https://opensource.hcltechsw.com/domino-c-api-docs

– https://learn.microsoft.com/en-us/windows/win32/apiindex/windows-api-list

– https://learn.microsoft.com/en-us/windows/win32/winprog/windows-data-types

• Also, either the code doesn’t work 
or it crashes your client

https://opensource.hcltechsw.com/domino-c-api-docs
https://learn.microsoft.com/en-us/windows/win32/apiindex/windows-api-list
https://learn.microsoft.com/en-us/windows/win32/winprog/windows-data-types


Notes API

• NSPingServer

• ODSReadMemory

• OSLock

• OSLockBlock

• OSLockObject

• OSMemoryLock

• OSTranslate (if using pointers)

Windows API

• lstrcpyA / lstrcpyW

• RtlMoveMemory

• GetActiveWindow

• Get/SetClipboardData

• Anything with “Handle”
– GetModuleHandle, CloseHandle…

• Anything with “Process”
– OpenProcess, GetCurrentProcess…

• Anything with “Service”
– OpenService, QueryServiceStatus…

Some Common Calls to Check for



Special Handling: Types/Structs with Strings

• Be careful if you use custom Types that have Strings
– Types map to structs in C/C++

– String parameters in function calls are fine, it’s just Strings inside Types here

Option Public
Option Declare

Public Type testType
s As String
i As Integer
l As Long

End Type

Sub Initialize
'** If you compile this on a 32-bit client, Len(t) will be 10 on both a 32-bit and 64-bit client. 
'** If you compile this on a 64-bit client, Len(t) will be 14 on both a 32-bit and 64-bit client. 
Dim t As testType
Print "testType size is " & Len(t)

End Sub



Special Handling: Types/Structs with Strings

• Size of the Type will depend on which client you compile on
– If you compile on 32-bit client, the String params will be 4 bytes

– If you compile on 64-bit client, the String params will be 8 bytes

• Can cause alignment problems with C-API calls

• Can cause problems if you have to pass the size of the struct
– Windows API sometimes does this as a struct versioning technique

• Kind of an edge case, not very common to have Strings in C-API Types
– Usually Strings are parameters, or structs contain pointers



Special Handling: Types/Structs with Strings

• Option 1: use pointers (Long/Double) instead of Strings
– Only works for Strings being returned, not Strings being passed

– Use OSTranslate or lstrcpy to convert between pointers and Strings

Declare Function pointerToString_64 Lib "kernel32" Alias "lstrcpyA" (_
ByVal toString As String, _
ByVal fromPointer As Double) As Long

Type testType_64
ptr As Double    '** String pointer, use Long for 32-bit

End Type

Declare Sub MyFunction_64 Lib "mydll" Alias “MyFunction" (tt As testType_64)

Sub Initialize
Dim tt As testType_64
Dim s As String*256

'** don't forget NotesSession.UseDoubleAsPointer before and after
Call MyFunction_64(tt)
Call pointerToString_64(s, tt.ptr)

Print "s = " & s
End Sub



Special Handling: Types /Structs with Strings

• Option 2: use Execute to dynamically compile and run
– Pass data to/from Execute block using public global variables

Public stringToPass As String
Public functionResult As Integer

Sub Initialize
stringToPass = "a beautiful string"

Execute |
Type testType

s As String
End Type

Declare Function SendTypeValue Lib "mydll" (tt As testType) As Integer

Sub Initialize
Dim tt As testType
tt.s = stringToPass
functionResult = SendTypeValue(tt)

End Sub |

Print "Result = " & functionResult
End Sub



Make Your Data Work For You

Java

:34



32-bit vs 64-bit Java

• In general, Java code that ran on 32-bit should work on 64-bit
– The JVM handles all the core language/platform things

– Be careful if you use JNI

– Search your code for System.loadLibrary calls

• Don’t forget to test your external/attached JAR files!
– They might make loadLibrary/DLL calls too



A History of Java in Notes/Domino



New in Notes 14: Java 17

• Notes had Java 8 since 9.0.1 FP8
– Lambdas, streams, most “modern” Java language features

– Still supported through at least November 2026, but getting old

– https://adoptium.net/support

• Notes 14 has Java 17
– A few new language features

– Many third party JAR files now require at least Java 11, if not 17

– Mostly backwards compatible, with a few exceptions…

https://adoptium.net/support


Missing in Java 17

• Java EE classes: javax.xml.ws, javax.xml.soap, javax.xml.bind, and more
– DatatypeConverter often used for hex <-> binary and base64 conversions

• sun.misc.BASE64Encoder / Decoder
– Also, com.ibm.misc.BASE64Encoder/Decoder has been gone since Notes 11

– Use java.util.Base64 classes instead (available since Java 8)

• com.sun.image.codec.jpeg.*
– Use javax.ImageIO (available since Java 1.4)

• Nashorn (or any) JavaScript engine
– e.g. ScriptEngineManager mgr = new ScriptEngineManager(); … 

– No native replacement, maybe use GraalVM?



New going from Java 8 -> 17

• Awesome new NullPointerException messages!
– They actually tell you what’s null

• Interesting language enhancements
– Switch statement pattern matching

– Record data type

– “var” keyword

– Text blocks (easy multi-line text)

– HttpClient class

• Java Migration Guide
– https://docs.oracle.com/en/java/javase/17/migrate/getting-started.html

https://docs.oracle.com/en/java/javase/17/migrate/getting-started.html


Java 17 jvm Folder

• Files in the jvm folder are different
– No more rt.jar and tools.jar

– Now just one big modules file

• Notes/jvm/lib/ext folder is now Notes/ndext
– Where external JAR files live

– notes.jar is there now too

• java.policy and java.security moved 
– Used to be notes/jvm/lib/security

– Now in notes/jvm/conf/security

– cacerts is still in notes/jvm/lib/security though



JAVA COMPILER SETTINGS



JAVA COMPILER SETTINGS

• In Domino Designer
– File – Preferences – Java – Compiler

• Default in Notes 14 is Java 17

• XPages compiled as Java 17 WILL NOT WORK on older versions of 
Notes client or Domino server
– Error 500 HTTP Web Server: Command Not Handled Exception

– domino\data\domino\workspace\logs\trace-log-0.xml : 

• java.lang.UnsupportedClassVersionError: JVMCFRE199E bad major version 61.0 of 

class=xsp/MyXPage, the maximum supported major version is 52.0;



JAVA COMPILER SETTINGS

• If you have pre-Notes 14 clients or servers using XPages, ALL 
DEVELOPERS should change compliance level to 1.8

• Not specifically a 64-bit thing, a Notes 14 thing



XPages Build Settings

• If you’re extra paranoid, turn Project – Build Automatically OFF
– So you don’t accidentally recompile XPages

– Sometimes all XPages in a database will recompile just by opening the database in 
Domino Designer if this setting is on

– https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0107841

https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0107841


BONUS: Common Domino Designer Problems

• Java code editor doesn't find Java classes with content assist (control-
space), and you have to enter import statements by hand
– FIX: right-click database in Designer and choose "Remove...“

– Then reopen the database in Designer

• Design element properties panel gets stuck to the right sidebar instead 
of below the agent code
– FIX: Window - Reset Perspective

• Code doesn’t have line numbers
– FIX: right-click the left “gutter” in the code window 

and choose “Show line numbers”

– Java and LotusScript



Upgrade Your Time

Eclipse Plugins

:42



64-bit Notes Client Has 64-bit Eclipse

• Obviously

• In general, you shouldn’t notice a difference

• Third-party plugins only a problem if they specifically target x86
– Eclipse-PlatformFilter or Bundle-NativeCode in MANIFEST.MF of plugin

– “arch” setting for <plugin> entry in feature.xml

• Unusual for plugins to specify x86
– Usually only if they make native/DLL calls and use JNI

– Could be the case with VoIP plugins, for example



x86 Plugins

• How can I tell?
– No easy way using standard Eclipse tricks

– Look for “x86” in MANIFEST.MF in plugin.jar and feature.xml in feature.jar

– Import into Update Site database and look at plugin and feature docs

• What if I use/install an x86 plugin on a 64-bit client?
– Eclipse won’t even load it

– After install there might be a message in the trace log, but often nothing

• How do I make it work?
– Ask the vendor for an x64 version



PowerShell Script To Search for x86 (rough cut)

function SearchInFiles ([string]$Path) {
Add-Type -AssemblyName System.IO.Compression.FileSystem

# search in jar files (which are really zip files)
Get-ChildItem -Path $Path -Filter "*.jar" | ForEach-Object {

$zipFile = $_.FullName
$zipArchive = [System.IO.Compression.ZipFile]::OpenRead($zipFile)
$zipArchive.Entries | Where-Object {($_.Name -eq "feature.xml") -or ($_.Name -eq "MANIFEST.MF")} | ForEach-Object {

$stream = $_.Open()
$reader = New-Object System.IO.StreamReader($stream)
while ($line = $reader.ReadLine()) {

if (($line -like "*arch=x86*") -or ($line -like "*arch=?x86*")) {
Write-Host "$zipFile\($_):$line"

}
}
$reader.Dispose()
$stream.Dispose()

}
$zipArchive.Dispose()

}

# search uncompressed features and plugins
ls -r -path $Path -filter feature.xml | select-string arch=`"x86`"
ls -r -path $Path -filter MANIFEST.MF | select-string arch=x86

}

# you could search in Notes\framework too, but ignore the Notes client plugins
SearchInFiles("C:\HCL\Notes\Data\workspace\applications\eclipse\plugins")
SearchInFiles("C:\HCL\Notes\Data\workspace\applications\eclipse\features")



SHA-1 Signatures No Longer Supported in Notes 14

• Plugins with SHA-1 signatures are “unsigned” in Notes 14
– Not specifically a 64-bit thing, a new Java restriction

– https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0109288

– Can still install plugins with old SHA-1 signatures if “Unsigned Plugin” is allowed

– Currently installed plugins with old signatures continue to work

• Re-sign your plugins with SHA-256 certificates if you need to

• Check META-INF/MANIFEST.MF in your old plugins if you’re curious
– Search for “SHA1-Digest”

– If you use the Update Site database, MANIFEST.MF is in the plugin and feature 
documents

https://support.hcltechsw.com/csm?id=kb_article&sysparm_article=KB0109288


UpdateSite Database On 12.0.2 FP1/2 64-bit

• Can’t import plugins to Notes UpdateSite database using 12.0.2 FP1/2 
64-bit client
– Problem on 64-bit Mac client for years, fixed in Notes 12.0.2 FP3 and 14.0.0

– For 12.0.2 FP1/2: use a 32-bit client or the OpenNTF version of the template at 
https://www.openntf.org/main.nsf/project.xsp?r=project/Open%20Eclipse%20Update%20Site

https://www.openntf.org/main.nsf/project.xsp?r=project/Open%20Eclipse%20Update%20Site


Upgrade Your Time

A Few Recommendations



A Few Recommendations

• If you’re upgrading to 64-bit, prefer Notes 14 to 12.0.2
– Many things got fixed, longer term support
– https://www.panagenda.com/webinars/how-to-perform-hcl-notes-14-upgrades-smoothly

• Be very careful compiling LotusScript in mixed 32/64 environments
– GetAllEntriesByKey and Replace bugs are not obvious!

• Be very careful compiling XPages using Notes 14
– Unless you change the Java compiler option to Java 1.8, or all your servers are v14

• Test and re-test your Notes API code

• Use the new HCL Download Portal for software downloads
– So much better than FlexNet -> https://my.hcltechsw.com

https://www.panagenda.com/webinars/how-to-perform-hcl-notes-14-upgrades-smoothly
https://my.hcltechsw.com/


A Collection of Links

https://www.panagenda.com/kbase/display/mc/Useful+HCL+Links

https://www.panagenda.com/kbase/display/mc/Useful+HCL+Links


Your Feedback Matters to Us

THANK YOU!THANK YOU!

Q & A

